

SASプロシジャを用いた生存時間データに対する例数設計の変革

○魚住 龍史¹ 矢田 真城² 浜田 知久馬³

¹京都大学大学院医学研究科 医学統計生物情報学

²エイツーヘルスケア株式会社 開発戦略本部 生物統計部

³東京理科大学 工学部 情報工学科

Revolution in sample size calculation for time-to-event data using SAS procedures

Ryuji Uozumi¹, Shinjo Yada², and Chikuma Hamada³

¹*Department of Biomedical Statistics and Bioinformatics, Kyoto University Graduate School of Medicine*

²*Biostatistics Department, Development Strategy Division, A2 Healthcare Corporation*

³*Department of Information and Computer Technology, Faculty of Engineering, Tokyo University of Science*

要旨

POWER プロシジャは、様々な解析手法に対する例数設計や検出力を計算するために有用なプロシジャであり、TWOSAMPLESURVIVAL ステートメントを用いれば、生存時間解析のノンパラメトリック検定を前提とした例数設計を行うことができる。TWOSAMPLESURVIVAL ステートメントでは、Lakatos の方法が採用されており、生存時間解析の例数設計としてよく用いられている Freedman の方法や Schoenfeld の方法と比べて性能が良いことが報告されている。その一方、中間解析を含んだ群逐次デザインに用いられる SEQDESIGN プロシジャでは、ログランク検定を前提とした例数設計法として Schoenfeld の方法が採用されている。また、SAS/STAT 14.1 からの POWER プロシジャでは、新たに COXREG ステートメントが追加され、連続量データの共変量を含んだ多変量の比例ハザードモデルの解析に対する例数設計ができるようになった。本稿では、SAS で新たにできるようになった生存時間解析の例数設計に関する機能を取りあげ、それらの有用性について報告する。まず、Lakatos の方法から算出される必要例数以外の出力結果として、必要イベント数や単位時間あたりの登録例数がどのように算出されているか解説する。次に、POWER プロシジャの COXREG ステートメントで実施できるようになった例数設計法の特徴と有用性を報告する。最後に、患者の登録パターンが一様でない場合の例数設計として、POWER プロシジャの代わりに、SEQDESIGN プロシジャで例数設計を行う方法を取りあげる。

キーワード：生存時間解析 例数設計 POWER TWOSAMPLESURVIVAL Lakatos COXREG Hsieh-Lavori SEQDESIGN Schoenfeld 登録パターン 切断指数分布

1 はじめに

医学研究では、ある基準の時刻から目的のイベントが起こるまでの時間を対象とした解析がよく行われる。このような解析手法を生存時間解析と呼び、LIFETEST, PHREG, LIFEREG といったプロシジャが 20 年以上用いられている (大橋・浜田, 1995)。近年では、SAS/STAT の機能拡張により、区間打ち切りデータに関するプロシジャとして、SAS/STAT 13.1 から ICLIFETEST, SAS/STAT 13.2 から ICPHREG が追加された (SAS Institute Inc., 2013, 2014)。さらに、SAS/STAT 14.1 からの LIFETEST プロシジャでは、EVENTCODE = オプションが追加され、競合リスクの解析が実施できるようになった (SAS Institute Inc., 2015)。

このように生存時間解析に関連したプロシジャが拡張されている一方、近年の医学研究では、研究目的を明確にするために、試験の必要例数を計画段階で事前に見積もらなければならない。これを例数設計と呼び、生存時間解析の例数設計を行うためには、POWER プロシジャにおける TWOSAMPLESURVIVAL ステートメントが有用である (浜田・安藤, 2005)。生存時間解析の例数設計法として、Freedman の方法 (Freedman, 1982) と Schoenfeld の方法 (Schoenfeld, 1981) がよく用いられているが、POWER プロシジャで採用されている Lakatos の方法 (Lakatos, 1988) の方が良い性能であることが報告されている (Lakatos and Lan, 1992; 中西ら, 2006; 水澤・浜田, 2008; 魚住ら, 2009)。また、SAS/STAT 14.1 からの POWER プロシジャでは、新たに COXREG ステートメントが追加され、連続量データの共変量を含んだ多変量の比例ハザードモデルの解析に対する例数設計ができるようになった (SAS Institute Inc., 2015)。

本稿では、SAS で新たにできるようになった生存時間解析の例数設計に関連した機能を取りあげる。まず、POWER プロシジャの TWOSAMPLESURVIVAL ステートメントによる Lakatos の方法から算出される必要イベント数や登録例数の算出について取りあげ、その有用性を示す。次に、POWER プロシジャの COXREG ステートメントで実施できるようになった例数設計法について解説し、TWOSAMPLESURVIVAL ステートメントと比較した場合の有用性を示す。最後に、POWER プロシジャの代わりに、SEQDESIGN プロシジャで生存時間解析の例数設計を行う有用性について取りあげる。

2 生存時間解析における例数設計法

生存時間解析においてはイベント数が情報量となるため、必要例数を求めるためには、まずは必要イベント数を見積もる必要がある。必要イベント数を求めた上で、

$$\text{必要例数} = \text{必要イベント数} / 2 \text{ 群を併せたイベントの割合} \quad (1)$$

を計算すればよい。

生存時間解析における必要イベント数は、Cox の比例ハザードモデルによるハザード比 (hazard ratio) を用いて算出される。比例ハザードモデルは、共変量ベクトル \mathbf{x} で条件付けたある患者集団のハザード関数 $h(t|\mathbf{x})$ が基準患者集団のハザード関数 $h_0(t)$ の定数倍になるモデルである (Cox, 1972)。ここで、回帰パラメータ ω 、単変量 x に対する比例ハザードモデル

$$\log[h(t|x)/h_0(t)] = \omega x \quad (2)$$

を考えると、ハザード比は

$$\frac{h(t|x=1)}{h(t|x=0)} = \exp(\omega)$$

となる。ただし、 t は時間を表す。

2.1 Freedman の方法と Schoenfeld の方法

生存時間解析の例数設計法として、Freedman の方法 (Freedman, 1982) と Schoenfeld の方法 (Schoenfeld, 1981) がよく用いられている。それぞれの方法で算出される必要イベント数を D_F , D_S とする。検定の有意水準を両側 α , 検出力の名義水準を $1 - \beta$ とすると、必要イベント数はそれぞれ次の式で求まる。

$$D_F = \frac{(z_{\alpha/2} + z_{\beta})^2 (w \cdot HR + 1)^2}{w(HR - 1)^2} \quad (3)$$

$$D_S = \frac{(1 + w)^2 (z_{\alpha/2} + z_{\beta})^2}{w \{\log(HR)\}^2} \quad (4)$$

ただし、 z_{κ} を標準正規分布の上側 $100 \cdot \kappa$ % 点、 HR をハザード比、 w を群間の例数の割付比 (対照群 : 試験群 = $1 : w$) とする。いずれの式においても 2 群を併せた必要イベント数を表しており、1 群あたりの必要イベント数を求めるためには、それぞれの式から求めた D_F あるいは D_S を 0.5 倍すればよい。(3), (4) 式のいずれも、生存時間分布に指数分布を仮定して導出される。

これらの方法は、テイラー展開により 2 次式で近似すると一致するため、帰無仮説 $H_0 : HR = 1$ の近傍では、両式はほぼ等しくなる (浜田・藤井, 2003)。また、 $w = 1$ の場合、Freedman の方法と比べて、Schoenfeld の方法の方が必要イベント数は必ず少なく済む。なお、いずれの方法においても、ハザード比を用いて例数設計を行っているが、ノンパラメトリックなログランク検定の例数設計とほぼ等価である。

ここで、試験群 (E) と対照群 (C) で構成される 2 群間比較試験を想定し、 $w = 1$ の場合で考える。このとき、必要例数を求めるために必要な条件となる、(1) 式の分母として示した 2 群を併せたイベントの割合について、各群における最終生存割合 π_C , π_E を用いて、

$$2 \text{ 群を併せたイベントの割合} = \frac{(1 - \pi_C) + (1 - \pi_E)}{2}$$

と見積もることができる。しかし、実際の臨床試験では、患者は逐次的に登録され、一度にすべての患者が組み入れられるわけではないため、登録期間を考慮する必要があるといえる。臨床試験の登録期間を A 年、フォローアップ期間を F 年として、範囲 $0 \sim A$ を確率密度 $1/A$ で一様分布に従って登録されると仮定すると、群 $g \in \{C, E\}$ における試験期間中の期待イベント割合は

$$P_g = 1 - \frac{1}{A} \int_0^A S_g(A + F - t) dt$$

となる。ただし、 $S_g(A + F - t)$ は時点 $(A + F - t)$ における患者の生存関数である。生存時間分布が指数分布に従うと仮定すると、

$$P_g = 1 - \frac{1}{A} \int_0^A \exp\{-\lambda_g \cdot (A + F - t)\} dt = 1 - \frac{\exp(-\lambda_g F) \{1 - \exp(-\lambda_g A)\}}{A \lambda_g} \quad (5)$$

となる。ただし、 λ_g は群 $g \in \{C, E\}$ におけるハザードを表す指数分布の母数である。よって、2 群を併せた期待イベント割合 $(P_C + P_E)/2$ を用いて、必要例数を見積もることができる。

2.2 SEQDESIGN プロシジャによる Schoenfeld の方法

2.1 節の (3), (4) 式に基づく Freedman の方法および Schoenfeld の方法による必要イベント数、さらに必要例数の計算には、DATA ステップによる実装が容易にできる (大橋ら, 2016)。3 節で述べる POWER プロシジ

による例数設計のように、SAS のプロシジャで出力させたい場合、Schoenfeld の方法であれば SEQDESIGN プロシジャを活用することで、必要イベント数と必要例数を算出することができる。

SEQDESIGN プロシジャは、本来群逐次デザインの検討を行うために有用なプロシジャである (浜田, 2010)。SAMPLESIZE ステートメントを指定することで、群逐次デザインの例数設計も行うことができる。

SEQDESIGN プロシジャでは、DESIGN ステートメントにおいて、群逐次デザインにおけるステージの数 (すなわち、中間解析の回数) を指定する。NSTAGE=1 とすれば、中間解析なしの固定デザインに相当する。

プログラム 1 は、検定の有意水準を両側 α 、検出力の名義水準を $1-\beta$ のデザインを想定している。ALT=TWOSIDED で両側検定を指定しており、片側検定の場合には LOWER あるいは UPPER を指定する。DESIGN ステートメントにおいて NSTAGE=1 を指定した下で、SAMPLESIZE ステートメントで MODEL=TWOSAMPLESURVIVAL を指定すれば、2 群の生存時間データに対してログランク検定を実施することを前提とした、Schoenfeld の方法に基づく必要イベント数および必要例数を算出できる。なお、SAMPLESIZE ステートメントのオプションとして、NULLHAZARD= で対照群に対するハザードを指定し、ACCTIME= で登録期間、FOLTIME= でフォローアップ期間、WEIGHT= で群間の割付比 (対照群の例数 : 試験群の例数 = 1:w) を指定している。

プログラム 1 : SEQDESIGN プロシジャによる Schoenfeld の方法に基づく例数設計

```
proc seqdesign altref=-log(HR);  
  design nstages=1 beta= $\beta$  alpha= $\alpha$  alt=twosided;  
  samplesize model=twosamplesurvival  
              (nullhazard= $\lambda_c$   
              ACCTIME=A FOLTIME=F  
              weight=w);  
run;
```

・数値例

生存時間データに対してログランク検定による群間比較を行うことを想定し、有意水準を 5% (両側)、検出力の名義水準を $1-\beta=0.80$ 、対照群 (C) における 5 年生存割合を 65%、試験群 (E) における 5 年生存割合を 80% (フォローアップ期間はすべての個体で等しく 5 年とする)、割付比として対照群の例数 : 試験群の例数 = 1:1 の条件の下、例数設計を行う場合を考える。各群の 5 年生存割合より、生存時間分布に指数分布を仮定すると、各群のハザードはそれぞれ 0.0862, 0.0446 となる。このとき、ハザード比 HR は 0.518, 対数ハザード比は $\log(HR)=-0.6578$ となる。プログラム 2 は、上記の条件の下、SEQDESIGN プロシジャで Schoenfeld の例数設計を行うためのプログラムである。

プログラム 2 を実行させると、必要イベント数は 72.56 となり、必要例数は 227.61 例となる。なお、プログラム 2 をそのまま実行すると、対数ハザード比などの値における丸めの誤差より、必要イベント数および必要例数は上述した値とならない。付録 A に詳細なプログラムを示しており、DATA ステップで (4) 式を計算した値と一致することを確認できる。

なお、SAMPLESIZE ステートメントによって、必要イベント数と必要例数に加えて、Information 「情報量」が出力されるが、これは対数ハザード比の分散の逆数であり、

$$I = D_s \times \frac{w}{(1+w)^2}$$

で算出される値である。

プログラム 2: 数値例に対する Schoenfeld の方法に基づく例数設計

```
proc seqdesign altref=0.6578;
  design nstages=1 beta=0.20 alpha=0.05 alt=twosided;
  samplesize model=twosamplesurvival
              (nullhazard=0.0862
              ACCTIME=2 FOLTIME=5
              weight=1);
run;
```

3 POWER プロシジャによる例数設計

POWER プロシジャにおける生存時間解析の例数設計に関するステートメントとして、TWOSAMPLESURVIVAL と COXREG が挙げられる。TWOSAMPLESURVIVAL ステートメントでは、Lakatos の方法に基づき例数設計が実施される (Lakatos, 1988)。その一方、COXREG ステートメントは SAS/STAT 14.1 から追加された機能であり、Hsieh-Lavori の方法に基づく例数設計法である (Hsieh and Lavori, 2000)。

3.1 TWOSAMPLESURVIVAL ステートメント

POWER プロシジャの TWOSAMPLESURVIVAL ステートメントにおいて採用されている Lakatos の方法では、観察期間を小区間に分けて区間ごとに期待イベント数を算出する。Freedman の方法や Schoenfeld の方法では、群間のハザードあるいは対数ハザードに差がないという帰無仮説を検定する一方、Lakatos の方法では区間ごとに群間のハザードに差がないという帰無仮説を検定する。

TWOSAMPLE ステートメントによる例数設計として、比例ハザードモデルを仮定した方法を考える。試験群 (E) と対照群 (C) で構成される 2 群間比較試験を想定し、時点 t^* における群 $g \in \{C, E\}$ の生存割合 $S_g(t^*)$ から、ハザード比

$$HR = \frac{\log S_E(t^*)}{\log S_C(t^*)}$$

を推定し、例数設計を行う場合を考える。プログラム 3 は、群間の割付比が対照群 : 試験群 = 1 : w の下、POWER プロシジャによるログランク検定に基づく例数設計法である。CURVE オプションで両群の生存割合を指定しているが、片方の群のみ指定した上でハザード比 HAZARDRATIO = を指定してもよい。なお、各群の生存割合として、1 時点のみ指定しているため、比例ハザード性を仮定した例数設計が行われる。

さらに、バージョン 9.3 からの POWER プロシジャでは、検出力の名義水準を指定した下で、必要例数の他にも、表 1 に示す値も算出できるようになった (SAS Institute Inc., 2011)。例えば、必要イベント数を算出するためには、プログラム 3 における NTOTAL = の代わりに EVENTSTOTAL = を指定すればよい。

表 1: 必要例数以外に算出できる値

TWOSAMPLESURVIVAL ステートメントにおける指定	算出される数値の内容
EVENTSTOTAL EVENTTOTAL EETOTAL =	期待イベント数
ACCRUALRATETOTAL ARTOTAL =	単位時間あたりの 2 群の登録例数
ACCRUALRATEPERGROUP ACCRUALRATEPERG ARPERGROUP ARPERG =	単位時間あたりの 1 群あたりの登録例数

プログラム 3 のように、指数分布を想定した例数設計を行う場合、Lakatos の方法による必要イベント数 D_L は、必要例数 N_L 、群 $g \in \{C, E\}$ における試験期間中の期待イベント割合 P_C, P_E 、および群間の割付比 (対照群の例数 : 試験群の例数 = 1 : w) を用いて、

$$D_L = N_L \times \frac{P_C + wP_E}{1 + w}$$

と算出される。詳細は付録 B を参照されたい。

プログラム 3: POWER プロシジャによる比例ハザード性を仮定した例数設計

```
proc power;
  twosamplesurvival test=logrank
  curve("Control") = (t*):(SC)
  curve("Experimental") = (t*):(SE)
  groupsurvival = "Control" | "Experimental"
  groupweights = (1 w)
  accrualtime = A followuptime = F
  ntotal = .
  power = 1-β;
run;
```

・数値例

プログラム 4 は、2.2 節の数値例に対して、比例ハザード性を仮定した Lakatos の方法に基づく例数設計を行うためのプログラムである。必要イベント数は 74.35、必要例数は 233.23 例となる。1 群あたりの必要例数を算出したい場合は、NTOTAL = の代わりに NPERGROUP = を指定すればよい。NTOTAL = や NPERGROUP = では、小数点以下を切り上げた例数が出力されるため、プログラム 4 の右側に示した必要例数を求めるためのプログラムでは NFRACCTOTAL を指定し、小数点以下を切り上げる前の例数も出力している。NTOTAL = の代わりに、ACCRUALRATETOTAL = を指定すると、1 年あたりの登録例数が $N_L/A = 233.23/2 = 116.62$ と出力される。

プログラム 4: 数値例に対する比例ハザード性を仮定した Lakatos の方法に基づく例数設計

必要イベント数	必要例数
<pre> proc power; twosamplesurvival test=logrank curve("Control") = (5):(0.65) curve("Experimental") = (5):(0.80) groupsurvival = "Control" "Experimental" groupweights = (1 1) accrualtime = 2 followuptime = 5 eventstotal = . power = 0.8; run; </pre>	<pre> proc power; twosamplesurvival test=logrank curve("Control") = (5):(0.65) curve("Experimental") = (5):(0.80) groupsurvival = "Control" "Experimental" groupweights = (1 1) accrualtime = 2 followuptime = 5 ntotal = . nfractional power = 0.8; run; </pre>

3.2 COXREG ステートメント

POWER プロシジャにおける生存時間解析の例数設計に関するステートメントとしては、以前は TWOSAMPLESURVIVAL のみであった。しかし、SAS/STAT 14.1 から新たに COXREG ステートメントが追加された。COXREG ステートメントでは、Hsieh-Lavori の方法に基づく例数設計が行われ、多変量の比例ハザードモデルの説明変数として、連続変数を加えた場合の変数間の相関を考慮した例数設計を実施できる。

3.2.1 Hsieh-Lavori の方法

2 節の (2) 式に示した、単変量に対する Cox の比例ハザードモデルによるスコア検定統計量は次のようになる。

$$S^2 = \frac{\sum_{i \in D} (x_i - e_i)^2}{\sum_{i \in D} v_i}$$

ただし、 D はイベント発現データの集合を表す。さらに、リスク集合を R として、

$$e_i = \frac{\sum_{j \in R} x_j}{n_i},$$

$$v_i = \frac{\sum_{j \in R} (x_j - e_i)^2}{n_i}$$

とする。

ここで、

$$\eta_i = \frac{\sum_{j \in R} x_j \exp(\alpha x_j)}{\sum_{j \in R} \exp(\alpha x_j)}$$

とすると、スコア検定統計量の分子は、

$$\sum_{i \in D} (x_i - e_i) = \sum_{i \in D} (x_i - \eta_i) + \sum_{i \in D} (\eta_i - e_i)$$

と分解できる. このとき, 第 1 項は漸近的に

$$\sum_{i \in D} (x_i - \eta_i) \sim N(0, \sum_{i \in D} \sum_{j \in R} (x_j - \eta_i)^2 / n_i)$$

であり, $\omega \rightarrow 0$ のとき,

$$\eta_i \rightarrow \frac{\sum_{j \in R} x_j}{n_i} = e_i,$$

$$\sum_{i \in D} \sum_{j \in R} \frac{(x_j - \eta_i)^2}{n_i} \rightarrow \sum_{i \in D} \sum_{j \in R} \frac{(x_j - e_i)^2}{n_i} = \sum_{i \in D} v_i$$

となるため,

$$\sum_{i \in D} (x_i - \eta_i) \sim N(0, \sum_{i \in D} v_i)$$

となる.

また, $\omega = 0$ の周りでテイラー展開による 1 次近似を行うと,

$$\begin{aligned} \eta_i &\approx \frac{\sum_{j \in R} x_j}{n_i} + \omega \cdot \frac{\left\{ \sum_{j \in R} x_j^2 \exp(\omega x_j) \right\} \left\{ \sum_{j \in R} \exp(\omega x_j) \right\} - \left\{ \sum_{j \in R} x_j \exp(\omega x_j) \right\}^2}{\left\{ \sum_{j \in R} \exp(\omega x_j) \right\}^2} \\ &= e_i + \omega \cdot v_i \end{aligned}$$

となるため,

$$\sum_{i \in D} (\eta_i - e_i) \approx \omega \sum_{i \in D} v_i$$

となる.

以上より, D に含まれるデータ数が D_{HL} とすると,

$$S = \frac{\omega \sum_{i \in D} v_i}{\sqrt{\sum_{i \in D} v_i}} = \omega \sqrt{\sum_{i \in D} v_i} = \omega \sqrt{\sigma^2 D_{HL}}$$

と表すことができる. ただし, σ^2 は単変量 x の分散である. ここで, 推定すべき回帰パラメータ $\omega = \log(HR)$

として, 必要イベント数は

$$D_{HL} = \frac{(z_{\alpha/2} + z_{\beta})^2}{\sigma^2 \{\log(HR)\}^2} \quad (6)$$

と表すことができる. (4) 式の Schoenfeld の方法と同様に, $\omega = 0$ における漸近式である. (4) 式に比べて保守的な例数設計の式であるといえる.

次に, 多変量 $\mathbf{x} = (x_1 \ x_2 \ \dots \ x_k)$ に対する Cox の比例ハザードモデル

$$\log[h(t|\mathbf{x})/h_0(t)] = \omega'_1 x_1 + \omega'_2 x_2 + \dots + \omega'_k x_k \quad (7)$$

を考えると, パラメータの推定値の分散を用いて得られる VIF (variance inflation factor) は,

$$VIF = \frac{V_k[\hat{\omega}'_1]}{V_1[\hat{\omega}]} = \frac{1}{1 - R^2}$$

となる。ただし、 $V_k[\hat{\omega}'_1]$ は(7)式の k 変量のモデルから得られる $\hat{\omega}'_1$ の分散、 $V_1[\hat{\omega}]$ は(2)式の単変量のモデルから得られる $\hat{\omega}$ の分散、 R^2 は x_2, x_3, \dots, x_k に対する x_1 の回帰式から得られる寄与率である。検出力を名義水準に保つために、多変量の比例ハザードモデルによる解析に対する必要イベント数は

$$D'_{HL} = D_{HL} \cdot VIF = \frac{D_{HL}}{1 - R^2} \quad (8)$$

となる。

(6), (8)式に基づき計算される必要イベント数を用いて、(1)式から必要例数を見積もることができる。

3.2.2 COXREG ステートメントによる Hsieh-Lavori の方法に基づく例数設計

プログラム5は、POWER プロシジャのCOXREG ステートメントを用いて、Hsieh-Lavoriの方法による例数設計を行うためのプログラムである。必要イベント数の計算式である(6)式は、(3), (4)式と異なり、 σ^2 を指定しなければならない。さらに、必要イベント数の計算には影響しないが、寄与率 R^2 も指定することができ、 R^2 が大きいほど必要例数は多くなる。また、必要例数を求める場合には、EVENTPROB = でイベント割合を指定する必要がある。

プログラム5: POWER プロシジャによる Hsieh-Lavori の方法に基づく例数設計

必要イベント数	必要例数
<pre>proc power; coxreg hazardratio = HR rsquare = R² stddev = SD eventstotal = . power = 1-β; run;</pre>	<pre>proc power; coxreg hazardratio = HR rsquare = R² stddev = SD ntotal = . eventprob = イベント割合 power = 1-β; run;</pre>

*SAS/STAT 14.1 より実行可能

・数値例

COXREG ステートメントによる例数設計の数値例として、SAS/STAT PHREG プロシジャのマニュアルで取り扱われている、多発性骨髄腫のデータを用いる。このデータは、アルキル化剤で65人の患者を治療する試験から得られた死亡または打ち切りまでの生存時間データであり、このうち48人がイベント発生、17人が打ち切りであった。生存時間の共変量として、診断時の尿素窒素を表すBUNの対数をとった値(LogBUN)、白血球数を表すWBCの対数をとった値、ヘモグロビン値、血清カルシウム値などの9つが取りあげられており、9つの共変量のうち予後因子を特定することに関心がある。

ここで、LogBUNが興味のある共変量であると仮定すると、他の8つの共変量との関連をあることが考えられる。このとき、他の8つの共変量に対するLogBUNの回帰式から得られる寄与率 R^2 は0.1839、VIFで示すと1.225となり、LogBUNの標準偏差は0.3126となる。

多発性骨髄腫のデータの LogBUN に関心がある場合、有意水準 5% (片側)、検出力の名義水準 80% の下、対数ハザード比 1 の差を検出できるような例数設計を行うにあたって、まず必要イベント数を求めると、(6) 式より 64 例と算出される。(1) 式より、必要イベント数から必要例数を求めると、イベント割合が $1 - 17/65 = 0.738$ なので、 $64/0.738 = 86.67$ となる。よって、(8) 式より、多変量の比例ハザードモデルによる解析に対する必要例数は、86.67 を $VIF = 1.225$ 倍して、106.17 となる。

プログラム 6 は、これらの計算を POWER プロシジャで実施するためのプログラムである。 R^2 を考慮して計算すると、必要イベント数は 78、必要例数は 105 と計算される。なお、 $R^2 = 0$ とすれば、必要イベント数が 64、必要例数が 86 と計算されることを確認できる。

プログラム 6：数値例に対する Hsieh-Lavori の方法に基づく例数設計

必要イベント数	必要例数
<pre>proc power; coxreg alpha = 0.05 sides = 1 hazardratio = 2.718 rsquare = 0.1837 stddev = 0.3126 eventstotal = . nfrac power = .8; run;</pre>	<pre>proc power; coxreg alpha = 0.05 sides = 1 hazardratio = 2.718 rsquare = 0.1837 stddev = 0.3126 ntotal = . nfrac eventprob = 0.7385 power = .8; run;</pre>

*SAS/STAT 14.1 より実行可能

プログラム 6 は、SAS/STAT14.1 から実行可能である。なお、日本語版の SAS/STAT 14.1 から、POWER プロシジャの出力結果は一部日本語で出力される。出力結果の詳細は付録 E を参照されたい。

4 登録パターンが一様でない場合の例数設計

3.1 節の POWER プロシジャにおける TWOSAMPLESURVIVAL ステートメントでは、脱落までの時間が指数分布に従うと仮定して、例数設計を行うことが可能である (大橋ら, 2016)。しかし、登録パターンについては、登録期間中に一様に患者が登録されると仮定した場合のみで、脱落までの時間のように指数分布を仮定した検討を行うことはできない。

しかし、2.2 節で取りあげた SEQDESIGN プロシジャを用いれば、期間内で一様に登録されるパターン以外に、切断指数 (truncated exponential) 分布に従って登録されると仮定することができる。切断指数分布の確率密度関数および累積分布関数はそれぞれ

$$g(t) = \frac{\gamma \exp(-\gamma t)}{1 - \exp(-\gamma A)},$$

$$G(t) = \frac{1 - \exp(-\gamma t)}{1 - \exp(-\gamma A)},$$

となる ($0 \leq t \leq A$). ただし, γ は形状パラメータである ($\gamma \neq 0$). 切断指数分布の累積分布関数の形状は図 1 のようになり (作成プログラムは付録 F を参照), 点線で示した一様分布と比較すると, $\gamma < 0$ のときは凸形の登録パターンとなり, その一方 $\gamma > 0$ のときは凹形の登録パターンとなる.

一様分布に従って登録される場合の期待イベント割合を表す (5) 式に対して, 登録パターンが切断指数分布に従う場合の期待イベント割合は,

$$P_g' = \int_F^{A+F} g(A+F-t) \{1 - \exp(-\lambda_g t)\} dt = 1 + \frac{\gamma \exp\{-\lambda_g(A+F)\} [1 - \exp\{(\lambda_g + \gamma)A\}]}{(1 - \exp(-\gamma A))(\lambda_g - \gamma)} \quad (9)$$

となる (Lachin and Foulkes, 1986). よって, $w=1$ の場合で考えると, 2.1 節と同様の手順, すなわち 2 群を併せた期待イベント割合 $(P_C' + P_E')/2$ を用いて, 必要例数を見積もることができる.




図 1: 切断指数分布の累積分布関数

4.1 SEQDESIGN プロシジャによる登録パターンが一様でない場合の例数設計

2.2 節に示した方法に基づき SEQDESIGN プロシジャを活用すれば, POWER プロシジャでは検討できなかった登録パターンが一様でない場合の例数設計を行うことができる. プログラム 7 は, SEQDESIGN プロシジャを用いて, 登録パターンが切断指数分布に従うと仮定した下で, 必要イベント数および必要例数を算出するためのプログラムである. 2.2 節同様, SEQDESIGN プロシジャの例数設計は Schoenfeld の方法に基づき実施され, 登録パターンの考慮についてのみ Lachin-Foulkes の方法に基づいている.

プログラム 7 では, SEQDESIGN プロシジャの SAMPLESIZE ステートメントにおいて, ACCRUAL = EXP(PARM = GAMMA') を指定することで, 登録パターンにパラメータ γ' の切断指数分布を仮定することができる. 注意すべき点として, ACCRUAL = で指定する GAMMA' は, (9) 式の γ を直接指定するのではなく, $\gamma' = A\gamma$ を指定する. 例えば, 登録期間として ACCTIME = 2 を指定しており, 図 1 で考えた $\gamma = -2$ の切断

指数分布を仮定したい場合には、`PARM = -4` と指定すればよい。なお、`SAMPLESIZE` ステートメントにおいて `ACCRUAL =` を指定しなければ、デフォルトである `ACCRUAL = UNIFORM` が適用される。

プログラム 7: SEQDESIGN プロシジャによる登録パターンが切斷指数分布に従うと仮定した例数設計

```
proc seqdesign altref=-log(HR);
  design nstages=1 beta= $\beta$  alpha= $\alpha$  alt=twosided;
  samplesize model=twosamplesurvival
              (nullhazard= $\lambda_c$ 
              ACCTIME=A FOLTIME=F
              accrual=exp(parm= $\gamma'$ )
              weight=w);
run;
```

・数値例

登録パターンが一様でない場合の数値例として、2.2 節の数値例と同様の条件であるが、図 1 のパラメータが $\gamma = -2, 2$ の切斷指数分布に登録パターンが従う場合を考える。

プログラム 8: 数値例に対する登録パターンが切斷指数分布に従う場合の例数設計

凸形の登録パターンの場合 ($\gamma = -2$)	凹形の登録パターンの場合 ($\gamma = 2$)
<pre>proc seqdesign altref=0.6578; design nstages=1 beta=0.20 alpha=0.05 alt=twosided; samplesize model=twosamplesurvival (nullhazard=0.08616 ACCTIME=2 FOLTIME=5 accrual=exp(parm=-4) weight=1); run;</pre>	<pre>proc seqdesign altref=0.6578; design nstages=1 beta=0.20 alpha=0.05 alt=twosided; samplesize model=twosamplesurvival (nullhazard=0.08616 ACCTIME=2 FOLTIME=5 accrual=exp(parm=4) weight=1); run;</pre>

プログラム 2 による実行結果より、必要イベント数は 72.56、必要例数は 227.61 例であった。SEQDESIGN プロシジャによる登録パターンが一様でない場合の例数設計では、(4) 式に基づき必要イベント数を求め、(9) 式から得られる期待イベント割合から、(1) 式に基づき必要例数が算出される。そのため、必要イベント数は登録パターンを変動させても同様の値が得られる。しかし、必要例数に関して、登録パターンが切斷指数分布に従う場合、プログラム 8 より、 $\gamma = -2$ の場合は 245.46 例に増大し、 $\gamma = 2$ の場合は 212.42 例に減少することを確認できる。

6 まとめ

本稿では、生存時間解析の例数設計に関連したプロシジャとして、POWER プロシジャ、SEQDESIGN プロシジャによる例数設計について取りあげた。

POWER プロシジャによる Lakatos の方法に基づく例数設計では、これまで例数のみしか出力できず、イベント数については算出できなかった。しかし、SAS/STAT 9.3 からイベント数についても算出され、例数設計法として性能が優れている Lakatos の方法を実務に適用する機会が増えることが予想される。

SAS/STAT 14.1 以降の POWER プロシジャでは、Lakatos の方法に加えて、Hsieh-Lavori の方法に基づく例数設計もできるようになった。がんを対象とした第 III 相試験のように、生存時間データを主要評価項目とした 2 群間比較試験の例数設計では、Lakatos の方法が有用である。しかし、疫学研究や非ランダム化比較試験では、生存時間に対する共変量として連続変数を加える場合があり、興味のある共変量が他の共変量と独立でないことが考えられる。このとき、Hsieh-Lavori の方法に基づき、VIF を考慮して例数設計を行うことが可能である。

また、POWER プロシジャでは、登録期間中に患者が一樣に登録されると仮定しており、登録パターンが一樣でない場合の検討を行うことができない。そこで、POWER プロシジャの代わりに、SEQDESIGN プロシジャで例数設計を行うことは有用であり、登録パターンが切断指数分布に従うと仮定した下で、Schoenfeld の方法に基づく例数設計を行うことができることを示した。Schoenfeld の方法は Lakatos の方法より性能が劣ってしまうが、生存時間解析の例数設計法として最もよく用いられている方法であり (Barthel et al, 2006)、日本語訳が出版されている生存時間解析のテキストにおいても取りあげられている (Collett, 2003; Hosmer et al., 2008)。そのため、実務上、登録パターンが一樣でない場合の必要イベント数、必要例数を検討するにあたって大変有用といえる。

比例ハザード性が成り立たない場合、比例ハザードモデルを仮定しない方法として、生存時間分布に区分直線モデルを仮定することで、精度良く例数設計を実施できる。POWER プロシジャにより、生存時間分布に区分直線モデルを仮定した例数設計を行うためには、CURVE オプションで 2 時点以上の生存割合を指定すればよい。比例ハザードモデルを仮定せずに、生存時間分布に区分直線モデルを仮定する方法については、魚住ら (2009)、大橋ら (2016) を参照されたい。

なお、POWER プロシジャによる Hsieh-Lavori の方法、SEQDESIGN プロシジャによる登録パターンが一樣でない場合の例数設計に関しては、DATA ステップによる実装も付録に示しており (付録 C, G)、プロシジャによる実行結果 (付録 D, H) と一致することを確認できる。

参考文献

- [1] Barthel FMS, Babiker A, Royston P, Parmar MK. Evaluation of sample size and power for multiarm survival trials allowing for non-uniform accrual, non-proportional hazards, loss to follow-up and cross-over. *Statistics in Medicine*. 25:2521–2542, 2006.
- [2] Collett D. *Modelling Survival Data in Medical Research* (2nd edn.). Chapman and Hall/CRC, Boca Raton, FL, 2003. 宮岡悦良(監訳), グラクソ・スミスクライン株式会社バイオメディカルデータサイエンス部: 安藤英一・今井由希子・遠藤輝・兼本典明・張方紅・寺尾工・橋本浩史・本間剛介(訳). 医薬統計のための生存時間データ解析. 共立出版, 2013.

- [3] Cox DR. Regression models and life-tables (with Discussion). *Journal of the Royal Statistical Society, Series B.* **34**:187–220, 1972.
- [4] Freedman LS. Tables of the number of patients required in clinical trials using the logrank test. *Statistics in Medicine.* **1**:121–129, 1982.
- [5] Johnson NL, Kotz S, Balakrishnan N. *Continuous Univariate Distributions, Vol.1* (2nd edn.). John Wiley and Sons, New York, NY, 1994.
- [6] Hosmer, D. W. Jr., Lemeshow, S. and May, S. *Applied Survival Analysis: Regression Modeling of Time to Event Data* (2nd edn.). John Wiley and Sons, Hoboken, NJ, 2008. 五所正彦(監訳), 佐藤泰憲・竹内久朗・長島健悟・中水流嘉臣・平川晃弘・松永信人・山田雅之(訳). 生存時間解析入門 (原書第2版). 東京大学出版会, 2014.
- [7] Hsieh FY, Lavori PW. Sample-size calculations for the Cox proportional hazards regression model with nonbinary covariates. *Controlled Clinical Trials.* **21**:552–560, 2000.
- [8] Lachin JM, Foulkes MA. Evaluation of sample size and power for analyses of survival with allowance for nonuniform patient entry, losses to follow-up, noncompliance, and stratification. *Biometrics.* **42**:507–519, 1986.
- [9] Lakatos E. Sample size based on the log-rank statistic in complex trials using the log-rank test. *Biometrics.* **44**:229–241, 1988.
- [10] Lakatos E, Lan KKG. A comparison of sample size methods for the logrank statistic. *Statistics in Medicine.* **11**:179–191, 1992.
- [11] SAS Institute Inc. *SAS/STAT(R) 9.3 User's Guide*. SAS Institute Inc., Cary, NC, 2011.
- [12] SAS Institute Inc. *SAS/STAT(R) 13.1 User's Guide*. SAS Institute Inc., Cary, NC, 2013.
- [13] SAS Institute Inc. *SAS/STAT(R) 13.2 User's Guide*. SAS Institute Inc., Cary, NC, 2014.
- [14] SAS Institute Inc. *SAS/STAT(R) 14.1 User's Guide*. SAS Institute Inc., Cary, NC, 2015.
- [15] Schoenfeld D. The asymptotic properties of nonparametric tests for comparing survival distribution. *Biometrika.* **68**:316–319, 1981.
- [16] 魚住龍史・水澤純基・浜田知久馬. 生存時間解析における Lakatos の症例数設計法の有用性の評価. SAS ユーザー総会論文集, 143–152, 2009.
- [17] 魚住龍史・森田智視. 生存時間解析における三種の神器. 呼吸, **34**: 1083–1089, 2015.
- [18] 大橋靖雄・浜田知久馬. 生存時間解析 –SAS による生物統計. 東京大学出版会, 1995.
- [19] 大橋靖雄・浜田知久馬・魚住龍史. 生存時間解析 応用編 –SAS による生物統計. 東京大学出版会, 2016.
- [20] 張方紅・寺尾工. 非劣性試験の症例数設計方法の紹介 –生存時間データの場合–. SAS ユーザー総会論文集, 87–100, 2010.
- [21] 張方紅. SAS による生存時間解析の実務. SAS ユーザー総会論文集, 167–184, 2011.
- [22] 中西豊支・五所正彦・菅波秀規. POWER プロシジャを用いた生存時間解析における症例数設定方法の統計学的一考察. SAS Forum ユーザー会論文集, 161–169, 2006.
- [23] 浜田知久馬・藤井陽介. 生存時間解析の症例数設計. 日本 SAS ユーザー会 論文集, 73–98, 2003.
- [24] 浜田知久馬・安藤英一. POWER プロシジャによる症例数設計. SAS Forum ユーザー会 論文集, 127–151, 2005.
- [25] 浜田知久馬. SAS による中間解析のデザインと解析. SAS ユーザー総会 論文集, 111–179, 2010.
- [26] 水澤純基・浜田知久馬. 生存時間解析における症例数設計方法の性能比較. SAS ユーザー総会 論文集, 19–28, 2008.

連絡先

E-mail : uozumi@kuhp.kyoto-u.ac.jp

付録 A : SEQDESIGN プロシジャによる Schoenfeld の方法に基づく例数設計の確認プログラム

```
*=====*;
* 追加の条件設定 ;

%let alpha=0.05; %let beta=0.2; %let power=%sysevalf(1-&beta); %let w=1;
%let Sc=0.65; %let Se=0.80;
%let HR=%sysevalf(%sysfunc(log(&Se))/%sysfunc(log(&Sc)));
%let logHR=%sysevalf(-%sysfunc(log(&HR)));
%let accrual=2; %let followup=5;
%let hc=%sysevalf(-%sysfunc(log(&Sc))/&followup);
%let he=%sysevalf(-%sysfunc(log(&Se))/&followup);
*=====*;
* SEQDESIGNプロシジャによるSchoenfeldの方法に基づく例数設計 ;

proc seqdesign altref=&logHR;
  design nstages=1 beta=&beta alpha=&alpha alt=twosided;
  sample size model=twosamplesurvival
    ( nullhazard=&hc weight=&w
      ACCTIME=&accrual FOLTIME=&followup accrual=uniform );
run;
*=====*;
* DATAステップによるSEQDESIGNプロシジャの計算結果の確認 ;

data sample;
  alpha=&alpha; beta=&beta; w=&w;
  za=probit(1-alpha/2); zb=probit(1-beta); Sc=&Sc; Se=&Se; HR=log(Se)/log(Sc);
  accrual=&accrual; followup=&followup; total=accrual+followup; hc=&hc; he=&he;
  Pc=1-exp(-hc*followup)*(1-(exp(-hc*accrual)))/(accrual*hc);
  Pe=1-exp(-he*followup)*(1-(exp(-he*accrual)))/(accrual*he);
  DtotalS=(1+w)**2*(za+zb)**2/(w*log(HR)**2);
  NtotalS=DtotalS*(1+w)/(Pc+w*Pe);
run;
```

付録 B : DATA ステップによる Lakatos の方法に基づく必要イベント数の計算プログラム

```
*=====*;
* Lakatosの方法に基づく必要例数 ;

ods output Output=lakatos;
proc power;
  twosamplesurvival test=logrank
  curve("Control") = (5):(0.65) curve("Experimental") = (5):(0.80)
  groupsurvival = "Control" | "Experimental"
  groupweights = (1 1) accrualtime = 2 followuptime = 5
  ntotal = . nfractional power = 0.8;
run;
*=====*;
* DATAステップによるLakatosの方法に基づく必要例数から必要イベント数の計算 ;

data lakatos; set lakatos;
  Sc=0.65; Se=0.80; HR=log(Se)/log(Sc);
  hc=-log(Sc)/FollowUpTime; he=-log(Se)/FollowUpTime;
  Pc=1-exp(-hc*FollowUpTime)*(1-(exp(-hc*AccrualTime)))/(AccrualTime*hc);
  Pe=1-exp(-he*FollowUpTime)*(1-(exp(-he*AccrualTime)))/(AccrualTime*he);
  Dtotal=FractionalN*(Weight1*Pc+Weight2*Pe)/(Weight1+Weight2);
run;
*=====*;
* Lakatosの方法に基づく必要イベント数の確認 ;
```

```

proc power;
  twosamplesurvival test=logrank
  curve("Control") = (5):(0.65) curve("Experimental") = (5):(0.80)
  groupsurvival = "Control" | "Experimental"
  groupweights = (1 1) accrualtime = 2 followuptime = 5
  eventstotal = . power = 0.8;
run;

```

付録 C : DATA ステップによる Hsieh-Lavori の方法に基づく必要イベント数の計算プログラム

```

*=====*;
* 変数LogBUNの標準偏差の確認 ;

proc means data=Myeloma;
  var logBUN;
run;
*=====*;
* VIFの確認 ;

proc reg data=Myeloma;
  model Time = LogBUN HGB Platelet Age LogWBC Frac LogPBM Protein SCalc / vif;
run;quit;
*=====*;
* 寄与率R2の確認 ;

proc reg data=Myeloma;
  model LogBUN = HGB Platelet Age LogWBC Frac LogPBM Protein SCalc;
run;quit;
*=====*;
* 条件の指定 ;

%let alpha=0.05; %let beta=0.2; %let power=%sysevalf(1-&beta); %let w=1;
%let logHR=1; %let HR=%sysfunc(exp(&logHR));
%let sigma=0.3126; %let VIF=1.225; %let rs=%sysevalf(1-1/&VIF);
%let n0=65; %let c0=17; %let eventp0=%sysevalf(1-&c0/&n0);
*=====*;
* DATAステップによるHsieh-Lavoriの方法に基づく例数設計 ;

data sample;
  alpha=&alpha; beta=&beta; w=&w; za=probit(1-alpha); zb=probit(1-beta);
  logHR=&logHR; HR=exp(logHR);
  sigma=&sigma; VIF=&VIF; rsquare=1-1/&VIF; pi=1-&c0/&n0;
  DtotalHL=(za+zb)**2/(sigma*logHR)**2; NtotalHL=DtotalHL/pi;
  NtotalHL_VIF=NtotalHL*VIF; DtotalHL_VIF=DtotalHL*VIF;
run;

```

付録 D : COXREG ステートメントによる Hsieh-Lavori の方法の計算プログラム

```

*=====*;
* 必要イベント数 - DtotalHL, DtotalHL_VIF に一致 ;

proc power;
  coxreg
  alpha = &alpha sides = 1 hazardratio = &HR rsquare = 0,&rs stddev = &sigma
  eventstotal = . nfrac power = &power;
run;
*=====*;
* 必要例数 - NtotalHL, NtotalHL_VIF に一致 ;

proc power;
  coxreg
  alpha = &alpha sides = 1
  hazardratio = &HR rsquare = 0,&rs stddev = &sigma
  ntotal = . nfrac eventprob = &eventp0 power = &power;
run;

```


付録 E: COXREG ステートメント (付録 D) による Hsieh-Lavori の方法の計算結果

● 必要イベント数

計算された Ceiling Event Total				
インデックス	R-square	Fractional Event Total	実際の検出力	Ceiling Event Total
1	0.000	63.268887	0.804	64
2	0.184	77.504387	0.802	78

● 必要例数

計算された上限の総標本サイズ				
インデックス	R-square	小数点の総標本サイズ	実際の検出力	上限の総標本サイズ
1	0.000	85.676618	0.801	86
2	0.184	104.953858	0.800	105

付録 F: 切断指数分布 (図 1) の作成プログラム

```

=====*;
* 切断指数分布に従う登録パターン (A = 2) ;
data trunexp;
a=2;
do gamma=-2,2; do t=0 to 2 by 0.01;
g=(1-exp(-gamma*t))/(1-exp(-gamma*a)); y=0.5*t; output;
end; end;
run;
=====*;
* 登録パターン別にグラフ化 ;
proc sgpanel data=trunexp noautolegend;
panelby gamma / columns=2;
series x=t y=g / lineattrs=(pattern=1 color=black);
series x=t y=y / lineattrs=(pattern=35 color=black);
rowaxis values=(0 to 1 by 0.1) label="G(t)";
colaxis values=(0 to 2 by 0.5) label="t";
run;

```

付録 G: DATA ステップによる登録パターンを考慮した例数設計プログラム

```

=====*;
* 付録Aの条件に加えて以下を設定 ;
%let gamma=-2; %let gamma seqdesign=%sysevalf(&accrual*&gamma);
=====*;
* DATAステップによる登録パターンを考慮したSchoenfeldの方法に基づく例数設計 ;
data sample;
alpha=&alpha; beta=&beta; w=&w; za=probit(1-alpha/2); zb=probit(1-beta);
gamma=&gamma; Sc=&Sc; Se=&Se; HR=log(Se)/log(Sc);
accrual=&accrual; followup=&followup; total=accrual+followup; hc=&hc; he=&he;
PcUni=1-exp(-hc*followup)*(1-(exp(-hc*accrual)))/(accrual*hc);
PeUni=1-exp(-he*followup)*(1-(exp(-he*accrual)))/(accrual*he);
PcExp=1+(gamma*exp(-hc*total)*(1-exp((hc-gamma)*accrual)))/((1-exp(-gamma*accrual))*(hc-gamma));
PeExp=1+(gamma*exp(-he*total)*(1-exp((he-gamma)*accrual)))/((1-exp(-gamma*accrual))*(he-gamma));
DtotsS=(1+w)**2*(za+zb)**2/(w*log(HR)**2);
NtotsS_Uni=DtotsS*(1+w)/(PcUni+w*PeUni);
NtotsS_Exp=DtotsS*(1+w)/(PcExp+w*PeExp);
run;

```

付録 H : SEQDESIGN プロシジャによる登録パターンを考慮した例数設計プログラム

```
*=====*;
* 登録パターンが一樣の場合の必要例数の確認 ;

proc seqdesign altnref=&logHR;
  design nstages=1 beta=&beta alpha=&alpha alt=twosided;
  samplesize model=twosamplesurvival
    ( nullhazard=&hc weight=&w
      ACCTIME=&accrual FOLTIME=&followup accrual=uniform );
run;
*=====*;
* 登録パターンが一樣でない場合の必要例数の確認 ;

proc seqdesign altnref=&logHR;
  design nstages=1 beta=&beta alpha=&alpha alt=twosided;
  samplesize model=twosamplesurvival
    ( nullhazard=&hc weight=&w
      ACCTIME=&accrual FOLTIME=&followup
      accrual=exp(parm=&gamma_seqdesign) );
run;
```